

CUBE

By Maurice McMullan

This program was written for the VZ-200 computer and requires a 16k expansion module. The program is a variation of one written by J. Schultz which was published in *Australian Personal Computer* 1982.

It allows the player to manipulate the Rubik's Cube by using various commands. The commands consist of a series

of instructions which rotate the sides of the cube in a clockwise direction through a number of right angles.

Special instructions permit:

1. Set up a random cube (to test the player's ability to solve the cube).
2. Store a cube on cassette.
3. Restore a cube from cassette.
4. To go back to the previous cube if

current instructions do not produce the desired effect.

5. If all else fails the program will arrive at a "solved" cube (by cheating of course).

A simple error detection routine determines if a side designator is incorrect and if so the command containing it is ignored.

```
4 CLS: PRINT@233, "***CUBE***";  
  
6 PRINT@291, "WRITTEN BY M.MC.MULLAN":FORA=1TO3000:NEXT  
  
7 CLS:PRINT@229, "INSTRUCTIONS?(YORN)";  
  
8 F$=INKEY$:K$=INKEY$:IFK$=""THEN8  
  
9 IFK$="Y"THENGOTO2000  
  
10 CLS:COLOR,1:CLEAR 420  
  
20 C$(1)="B":C$(2)="F":C$(3)="R":C$(4)="L":C$(5)="D":C$(6)="U"  
  
30 CD(1)=2:CD(2)=6:CD(3)=18:CD(4)=12:CD(5)=4:CD(6)=21
```

47

PERSONAL COMPUTER GAMES

V 2(7): Oct. 85
p 47-52.
1 of 6.


```

40 DIM I(9,6)
50 REM SET UP FOR PERFECT CUBE
55 FOR A=1 TO 9: FOR B=1 TO 6: I(A,B)=B: NEXT: NEXT
60 GOSUB 400: GOTO 720
100 REM SUBROUTINE TO TURN FACE
110 ITEMP=I(8,N): JTEMP=I(7,N)
120 FOR INC=6 TO 1 STEP -1: I(INC+2,N)=I(INC,N): NEXT
130 I(2,N)=ITEMP: I(1,N)=JTEMP: RETURN
200 REM SUBROUTINE TO CHANGE AN EDGE
210 FOR REP=1 TO GO: RESTORE
230 IF COM=1 THEN FORDUM=1 TO (COM-1)*24: READ SKIP: NEXT DUM
240 FOR I2NC=1 TO 3
250 READ PO, FA: ITEMP=I(PO, FA)
260 FOR INC=1 TO 3
270 READ P2, F2: I(P2, FA)=I(PO, F2)
280 PO=P2: FA=F2: NEXT INC: I(P2, F2)=ITEMP: NEXT I2NC
320 N=COM: GOSUB 100: NEXT REP: RETURN
400 REM SUBROUTINE FOR PRINTING CUBE
410 RESTORE
420 RESTORE: FORDUM=1 TO 144: READ SKIP: NEXT
430 FOR Y=0 TO 64 STEP 32: FOR X=29098 TO 29100: READ PO, FA
470 POKE X+Y, CD(I(PO, FA)): NEXT: NEXT
485 FOR A=0 TO 256 STEP 128: FOR Y=0 TO 64 STEP 32
495 FOR X=29034 TO 29036: READ PO, FA
505 POKE X-Y-A, CD(I(PO, FA)): NEXT: NEXT: NEXT
515 FOR A=0 TO 8 STEP 8: FOR Y=0 TO 64 STEP 32: FOR X=28966 TO 28968
530 READ PO, FA: POKE X+A+Y, CD(I(PO, FA)): NEXT: NEXT: NEXT: RETURN
600 REM SET UP INSTRUCTION TO GO BACK TO PREVIOUS SET UP
605 X$="/": U=0
610 U=U+2: A$=MID$(Y$, U-1, 2)

```



```

630 E$=LEFT$(A$,1):D$=MID$(A$,2,1)
650 J=4-(ASC(D$)-48)
660 G$=CHR$(J+48)
662 H$=E$+G$
664 X$=H$+X$
670 IFU+1=LEN(Y$)THENRETURNELSE610
720 PRINT@0,"";
725 X$="":PRINT@0,"ENTER COMMANDS ";:INPUTX$
727 LL=0
728 LL=LL+1:IFMID$(X$,LL,1)<>"/"THEN728
729 LL=LL-1
730 AA=0
740 AA=AA+2:Z$=MID$(X$,AA-1,2)
745 REM Q = END GAME
750 IFLEFT$(Z$,1)="Q"THENCLS:COLOR,0:END
755 REM Y = PERFECT CUBE
760 IFLEFT$(Z$,1)="Y"THEN50
765 REM I = RETURN TO LAST ATTEMPT
770 IFLEFT$(Z$,1)="I"THENGOSUB600:GOTO727
777 REM X= RANDOM CUBE
778 IFLEFT$(Z$,1)="X"THEN784
779 REM T = STORE CURRENT CUBE ON TAPE
780 IFLEFT$(Z$,1)="T"THEN960
781 REM P = RESTORE CUBE FROM TAPE
782 IFLEFT$(Z$,1)="P"THEN990
783 Y$=X$:GOTO820:REM SAVE CURRRENT CUBE
784 REM SET UP RANDOM CUBE
785 X$="F2":FORJ=1TO9
786 F=RND(5)+1:X$=X$+C$(INT(F))+CHR$(INT(RND(2)+49)):NEXT
787 X$=X$+"/":GOTO727

```



```

820 REM DETERMINE WHICH SIDE AND HOW FAR TO ROTATE
825 G=0:A=0
830 A=A+1:IFMID$(Z$,1,1)=C$(A)THENG=1
840 IFG=0ANDA<6THEN830
850 IFG=1THENCOM=A:GOTO870
860 JP=0:PRINT@0,"ERROR      IN  ";:GOSUB2500:IFJP=1THEN720
865 GOTO727
870 Z$=MID$(Z$,2,1)
890 IFASC(Z$)>=49ANDASC(Z$)<=51THENG=ASC(Z$)-48:GOTO910
895 IFASC(Z$)=48THEN930ELSEZ$=CHR$(ASC(Z$)-4):GOTO890
910 GOSUB200
920 GOSUB400
930 IFAA<LLTHEN740
940 GOTO720
960 REM STORE CUBE ON TAPE:
961 CLS:PRINT@166,"START TAPE TO RECORD";
962 PRINT@200,"PRESENT SOLUTION";
963 PRINT@259,"PRESS ANY KEY TO CONTINUE"
965 F$=INKEY$:D$=INKEY$:IFD$=""THEN965
970 A$="":FORA=1TO9:FORB=1TO6:A$=A$+CHR$(I(A,B)+48):NEXT:NEXT
975 PRINT#"TEMSOL",A$
980 Z$="Q":GOTO750
990 REM RESTORE CUBE FROM TAPE
991 CLS:PRINT@166,"START TAPE TO INPUT";
992 PRINT@200,"STORED SOLUTION";
993 PRINT@259,"PRESS ANY KEY TO CONTINUE"
994 F$=INKEY$:D$=INKEY$:IFD$=""THEN994
995 INPUT#"TEMSOL",A$
996 L=1:FORA=1TO9:FORB=1TO6:I(A,B)=ASC(MID$(A$,L,1))-48:L=L+1
997 NEXT:NEXT

```



```

998 CLS:GOSUB400:GOTO720

1000 REMDATA FOR MOVES

1010 DATA3,4,5,6,7,3,5,5,4,4,6,6,8,3,6,5,5,4,7,6,1,3,7,5
1020 DATA7,4,1,5,3,3,1,6,8,4,2,5,4,3,2,6,1,4,3,5,5,3,3,6
1030 DATA7,2,3,5,3,1,7,6,8,2,4,5,4,1,8,6,1,2,5,5,5,1,1,6
1040 DATA7,1,7,5,3,2,3,6,8,1,8,5,4,2,4,6,1,1,1,5,5,2,5,6
1050 DATA1,3,1,2,1,4,1,1,2,3,2,2,2,4,2,1,3,3,3,2,3,4,3,1
1060 DATA5,3,5,1,5,4,5,2,6,3,6,1,6,4,6,2,7,3,7,1,7,4,7,2

1100 REM DATA FOR PRINTING

1110 DATA1,5,2,5,3,5,8,5,9,5,4,5,7,5,6,5,5,5
1120 DATA3,2,2,2,1,2,4,2,9,2,8,2,5,2,6,2,7,2
1130 DATA3,6,2,6,1,6,4,6,9,6,8,6,5,6,6,6,7,6
1140 DATA7,1,6,1,5,1,8,1,9,1,4,1,1,1,2,1,3,1
1150 DATA5,4,6,4,7,4,4,4,9,4,8,4,3,4,2,4,1,4
1160 DATA5,3,6,3,7,3,4,3,9,3,8,3,3,3,2,3,1,3

2000 REM INSTRUCTIONS

2005 CLS:PRINT"***** INSTRUCTIONS *****":PRINT
2010 PRINT"THIS PROGRAM ALLOWS ONE TO PLAY"

2020 PRINT"WITH THE RUBIC CUBE"

2030 PRINT"SIDES ARE LETTERED: -"

2040 PRINT"          B          BACK"
2050 PRINT"          F          FRONT"
2060 PRINT"          U          UPPER"
2070 PRINT"          L          LEFT"
2080 PRINT"          R          RIGHT"
2090 PRINT"          D          DOWN"

2100 PRINT:PRINT"***PRESS <C> TO CONTINUE ***":PRINT
2110 F$=INKEY$:D$=INKEY$:IFD$<>"C"THEN2110
2120 CLS:PRINT"INSTRUCTIONS ARE ENTERED AS :-"
2130 PRINT" 1.ROTATION OF SIDES."

```



```

2140 PRINT"SIDES ARE ROTATED IN A CLOCKWISE"
2150 PRINT"DIRECTION THROUGH A NUMBER OF RIGHT"
2160 PRINT"ANGLES.THE SENSE OF THE ROTATION"
2170 PRINT"OF A FACE IS TAKEN WHEN ONE "
2180 PRINT"LOOKS DIRECTLY AT THAT FACE"
2184 PRINT:PRINT"**** PRESS <C> TO CONTINUE ***":PRINT
2185 F$=INKEY$:D$=INKEY$:IFD$<>"C"THEN2185
2190 CLS:PRINT" AN EXAMPLE OF AN INSTRUCTION IS"
2200 PRINT"          R2L3U1B3/":PRINT
2205 PRINT"  MUST END COMMANDS WITH A /":PRINT
2210 PRINT"THIS MEANS ROTATE:-"
2215 PRINT"  RIGHT FACE THROUGH 180DEG"
2220 PRINT"  LEFT FACE THROUGH 270DEG"
2225 PRINT"  UPPER FACE THROUGH 90DEG"
2230 PRINT"  BACK FACE THROUGH 270DEG"
2250 PRINT:PRINT"**** PRESS <C> TO CONTINUE ****":PRINT
2260 F$=INKEY$:D$=INKEY$:IFD$<>"C"THEN2260
2270 CLS:PRINT" 2. SPECIAL INSTRUCTIONS"
2280 PRINT"  Q      QUIT GAME"
2290 PRINT"  Y      SET UP PERFECT CUBE"
2300 PRINT"  I      RETURN TO LAST ATTEMPT"
2310 PRINT"  X      SET UP RANDOM CUBE"
2320 PRINT"  T      STORE CUBE ON TAPE"
2330 PRINT"  P      RESTORE CUBE FROM TAPE"
2333 PRINT:PRINT"****PRESS <C> TO CONTINUE****":PRINT
2334 F$=INKEY$:D$=INKEY$:IFD$<>"C"THEN2334
2340 GOTO10
2500 REM ERROR ROUTINE
2510 FORJ=1TOLLSTEP2
2520 IFMID$(X$,J,2)=Z$THEN2540
2530 NEXTJ:PRINT@0,"NO ERROR FOUND?":JP=1:RETURN
2540 IFJ=1THENJP=1:RETURN
2545 Y$=MID$(X$,1,J-1)+"/"
2550 GOSUB600:RETURN

```